
This package implements Low-Rank Adaptation (LoRA), a popular method for fine-tuning large language models.

  
    
  
  Installation


This package can be installed by adding lorax to your list of dependencies in mix.exs:
def deps do
  [
    {:lorax, "~> 0.1.0"}
  ]
end

  
    
  
  Fine-tuning an LLM with LoRA


In general,
	Import your model
	Inject trainable LoRA parameters
	Train LoRA model
	Download LoRA only params

{:ok, model_info} = Bumblebee.load_model({:hf, "gpt2"})
%{model: gpt2_model, params: gpt2_params} = model_info

lora_model =
  gpt2_model
  |> Axon.freeze()
  |> Lorax.inject(%Lorax.Config{
      r: 2,
      alpha: 4,
      dropout: 0.05
  })

lora_merged_params =
  Axon.build(lora_model, mode: :train)
  |> Axon.Loop.trainer(custom_loss_fn, Polaris.Optimizers.adam(learning_rate: 3.0e-4))
  |> Axon.Loop.run(train_batch_stream, gpt2_params, epochs: 3, iterations: 1000, compiler: EXLA)

lora_params = lora_merged_params
  |> Lorax.Params.filter(gpt2_params)
  |> Lorax.Params.kino_download()
In practice, every model has some unique architecture that you need to account for.
For more detailed guides, see
	Finetuning LLMs with LoRA
	Running LLMs with LoRA


  
    
  
  Default Settings


The default config applies LoRA to all query, key, value matrices. r = 1, alpha = 2.
The LoRA paper demonstrated that adapting only the query and value matrices with r = 1 achieved effective fine-tuning results. However, for larger language models, people often choose much higher values of r and sometimes target all linear layers.

  
    
  
  Recommended Settings


These settings works well for fine-tuning smaller LLMs (~1b param models)
Lora Config
- r value  = at least 2
- alpha value = r * 2

Training
- learning_rate of 3.0e-4 with an adam optimizer

Text Generation
- multinomial sampling
- p = 0.06 or 0.08 for more variety (or if you experience repetitive results)
For more details on configuring LoRA hyperparameters, see this post by Sebastian Raschka.

  
    
  
  Limitations


	GPU Memory Requirements: Although LoRA reduces the GPU requirements for fine-tuning, larger LLMs beyond GPT2 still demand GPUs with substantial vRAM. Inadequate memory management can lead to cuda OOM crashes.

	Fine-Tuning Speed: The training speed of this library isn't on par with Huggingface's PEFT library. Further optimizations can be done to close the gap.


Note: For minor fine-tuning tasks without a GPU, the BinaryBackend is a viable option, often resulting in smoother training runs. Future updates will focus on minimizing GPU memory usage by reducing the amount of tensors stored during training, and potentially a QLoRA implementation one day.



  

    
Finetuning LLMs with LoRA
    

Mix.install([
  {:bumblebee, "~> 0.4.2"},
  {:axon, "~> 0.6.0"},
  {:nx, "~> 0.6.1"},
  {:exla, "~> 0.6.1"},
  {:lorax, "~> 0.1.0"},
  {:req, "~> 0.4.0"},
  {:kino, "~> 0.11.0"}
])

Nx.default_backend(EXLA.Backend)

  
    
  
  Introduction


This notebook will show how to train a LoRA adapter w/ GPT2. Most of this notebook is copied from https://hexdocs.pm/bumblebee/fine_tuning.html. If you want to learn more about how this training setup works, check that livebook instead. The LoRA specific details are further down below.

  
    
  
  Hyperparameters


batch_size = 2
sequence_length = 256
r = 2
lora_alpha = 4
lora_dropout = 0.05

:ok
:ok

  
    
  
  Load a model


{:ok, spec} = Bumblebee.load_spec({:hf, "gpt2"})
{:ok, model} = Bumblebee.load_model({:hf, "gpt2"}, spec: spec)
{:ok, tokenizer} = Bumblebee.load_tokenizer({:hf, "gpt2"})
{:ok, generation_config} = Bumblebee.load_generation_config({:hf, "gpt2"})

:ok

17:24:05.365 [info] TfrtCpuClient created.

:ok

  
    
  
  Prepare a dataset


We'll be using some data scraped from Elixirforum.
It's structured like this.
<title>Ideas for an Open Source Discord</title>

<author>WildYorkies</author>

I remember seeing on the Elixir ...
<likes>3 likes</likes>
text =
  Req.get!("https://raw.githubusercontent.com/wtedw/lorax/main/data/elixir-discussion.txt").body

:ok
:ok
tokenized_text = %{"input_ids" => input_ids} = Bumblebee.apply_tokenizer(tokenizer, text)
n_tokens = Nx.size(input_ids)
n_train = round(n_tokens * 0.9)
n_val = n_tokens - n_train

train_data =
  for {input_key, tokenized_values} <- tokenized_text, into: %{} do
    {input_key, Nx.slice_along_axis(tokenized_values, 0, n_train, axis: -1)}
  end

test_data =
  for {input_key, tokenized_values} <- tokenized_text, into: %{} do
    {input_key, Nx.slice_along_axis(tokenized_values, n_train, n_val, axis: -1)}
  end
%{
  "attention_mask" => #Nx.Tensor<
    u32[1][383105]
    EXLA.Backend<host:0, 0.2032894552.1868431376.195217>
    [
      [1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, ...]
    ]
  >,
  "input_ids" => #Nx.Tensor<
    u32[1][383105]
    EXLA.Backend<host:0, 0.2032894552.1868431376.195218>
    [
      [952, 30120, 25, 198, 40, 447, 247, 76, 2111, 284, 2050, 922, 7572, 329, 2615, 42652, 9643, 6725, 13, 198, 40, 481, 467, 4622, 503, 286, 7243, 994, 13, 1002, 345, 447, 247, 260, 655, 3599, 503, 314, 561, 1950, 326, 345, 467, 329, 262, 10314, 26, 42652, ...]
    ]
  >,
  "token_type_ids" => #Nx.Tensor<
    u32[1][383105]
    EXLA.Backend<host:0, 0.2032894552.1868431376.195219>
    [
      [0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, ...]
    ]
  >
}
defmodule DataStream do
  def get_batch_stream(%{"input_ids" => input_ids} = data, batch_size, block_size, opts \\ []) do
    seed = Keyword.get(opts, :seed, 1337)

    Stream.resource(
      # initialization function
      fn ->
        Nx.Random.key(seed)
      end,
      # generation function
      fn key ->
        {_b, t} = Nx.shape(input_ids)

        data =
          for {k, v} <- data, into: %{} do
            {k, Nx.reshape(v, {t})}
          end

        # ix = list of random starting indices
        {ix, new_key} =
          Nx.Random.randint(key, 0, t - block_size, shape: {batch_size}, type: :u32)

        ix = Nx.to_list(ix)

        # x is map of sliced tensors
        x =
          for {k, tensor} <- data, into: %{} do
            batch_slice =
              ix
              |> Enum.map(fn i -> Nx.slice_along_axis(tensor, i, block_size, axis: -1) end)
              |> Nx.stack()

            {k, batch_slice}
          end

        # y represents all the predicted next tokens (input_ids shifted by 1) 
        y =
          ix
          |> Enum.map(fn i ->
            data["input_ids"] |> Nx.slice_along_axis(i + 1, block_size, axis: -1)
          end)
          |> Nx.stack()
          |> Nx.flatten()

        out_data = {x, y}

        {[out_data], new_key}
      end,
      fn _ -> :ok end
    )
  end
end
{:module, DataStream, <<70, 79, 82, 49, 0, 0, 16, ...>>, {:get_batch_stream, 4}}
You can see what a single batch looks like by grabbing 1 from the stream:
train_batch_stream = DataStream.get_batch_stream(train_data, batch_size, sequence_length)
test_batch_stream = DataStream.get_batch_stream(test_data, batch_size, sequence_length)

[{x, y}] = train_batch_stream |> Enum.take(1)
[{x_val, y_val}] = test_batch_stream |> Enum.take(1)

Bumblebee.Tokenizer.decode(tokenizer, x["input_ids"]) |> IO.inspect()
IO.puts("=====")
Bumblebee.Tokenizer.decode(tokenizer, y) |> IO.inspect()

:ok
[" to how Greg’s Event Store works with its competing consumers model. Why would you want to do this? It allows handlers to run at different speeds, typically you have slow async handlers that can lag behind (e.g. sending emails, third party API requets). But you don’t want them to hold up read model projections to minimise query latency.\nAutonomous subscriptions allows you to add new handlers and replay all events from the beginning of time, or restart a handler to rebuild a projection. I’ve implemented a hybrid push/pull model for the Event Store subscriptions where appended events are published to subscribers, but they are buffered per subscriber and use back-pressure to ensure the subscriber isn’t overwhelmed. The subscription falls back to pulling events from the store when it gets too far behind, until caught up again.\nYou could use GenStage for this, but I would recommend using an individual flow pipeline per handler; not one flow for all handlers. Since GenStage's broadcast dispatcher can only go as fast as the slowest consumer. You also want to have any event handlers run from the event store, after the events have been atomically persisted. Appending events to the store should guarantee that a success reply is returned",
 " by a more fundamental principle - of developer happiness and a system that ‘just makes sense’. Of course the syntax is itself inspired by this, but it goes beyond syntax.\nHaving said that, I think José has also tried to stay true to Erlang and this certainly shows when using Elixir.\nWhen would I use Ruby? When I don’t need Elixir When I write a script for the server, or need to put a site up quick, or have a smaller project in mind I would use Ruby. Mainly for two reasons: I know it, and there is a huge community/set of libraries out there. Chances are if you want to do something someone already has in Ruby.\nThat may well change as I learn Elixir. I’m hoping it does actually, as my brain can’t hold too much information so sticking to one language would be preferential for me\n\n<likes>2 likes</likes>\n\n<author>gnat</author>\n\nAstonJ:\nmy brain can’t hold too much information so sticking to one language would be preferential for me\nThat was at least part of where I was coming from in raising the original question. I used to code"]
=====
" how Greg’s Event Store works with its competing consumers model. Why would you want to do this? It allows handlers to run at different speeds, typically you have slow async handlers that can lag behind (e.g. sending emails, third party API requets). But you don’t want them to hold up read model projections to minimise query latency.\nAutonomous subscriptions allows you to add new handlers and replay all events from the beginning of time, or restart a handler to rebuild a projection. I’ve implemented a hybrid push/pull model for the Event Store subscriptions where appended events are published to subscribers, but they are buffered per subscriber and use back-pressure to ensure the subscriber isn’t overwhelmed. The subscription falls back to pulling events from the store when it gets too far behind, until caught up again.\nYou could use GenStage for this, but I would recommend using an individual flow pipeline per handler; not one flow for all handlers. Since GenStage's broadcast dispatcher can only go as fast as the slowest consumer. You also want to have any event handlers run from the event store, after the events have been atomically persisted. Appending events to the store should guarantee that a success reply is returned only a more fundamental principle - of developer happiness and a system that ‘just makes sense’. Of course the syntax is itself inspired by this, but it goes beyond syntax.\nHaving said that, I think José has also tried to stay true to Erlang and this certainly shows when using Elixir.\nWhen would I use Ruby? When I don’t need Elixir When I write a script for the server, or need to put a site up quick, or have a smaller project in mind I would use Ruby. Mainly for two reasons: I know it, and there is a huge community/set of libraries out there. Chances are if you want to do something someone already has in Ruby.\nThat may well change as I learn Elixir. I’m hoping it does actually, as my brain can’t hold too much information so sticking to one language would be preferential for me\n\n<likes>2 likes</likes>\n\n<author>gnat</author>\n\nAstonJ:\nmy brain can’t hold too much information so sticking to one language would be preferential for me\nThat was at least part of where I was coming from in raising the original question. I used to code pretty"
:ok

  
    
  
  Train the model


%{model: model, params: params} = model

model
#Axon<
  inputs: %{"attention_head_mask" => {12, 12}, "attention_mask" => {nil, nil}, "cache" => nil, "input_embeddings" => {nil, nil, 768}, "input_ids" => {nil, nil}, "position_ids" => {nil, nil}}
  outputs: "container_37"
  nodes: 859
>
[{input, _}] = Enum.take(train_batch_stream, 1)
Axon.get_output_shape(model, input)
%{
  cache: #Axon.None<...>,
  hidden_states: #Axon.None<...>,
  attentions: #Axon.None<...>,
  cross_attentions: #Axon.None<...>,
  logits: {2, 256, 50257}
}
We'll need to freeze the original parameters in our model before injecting. That way we can train only the LoRA nodes.
lora_model =
  model
  |> Axon.freeze()
  |> Lorax.inject(%Lorax.Config{
    r: r,
    alpha: lora_alpha,
    dropout: lora_dropout,
    target_key: true,
    target_query: true,
    target_value: true
  })
#Axon<
  inputs: %{"attention_head_mask" => {12, 12}, "attention_mask" => {nil, nil}, "cache" => nil, "input_embeddings" => {nil, nil, 768}, "input_ids" => {nil, nil}, "position_ids" => {nil, nil}}
  outputs: "container_37"
  nodes: 895
>
We'll reshape the usual GPT2 output so that we can use it with categorical_cross_entropy
defmodule CommonTrain do
  import Nx.Defn

  defn custom_predict_fn(model_predict_fn, params, input) do
    %{prediction: preds} = out = model_predict_fn.(params, input)

    # Output of GPT2 model is a map containing logits and other tensors
    logits = preds.logits

    {b, t, c} = Nx.shape(logits)
    reshaped = Nx.reshape(logits, {b * t, c})
    %{out | prediction: reshaped}
  end

  def custom_loss_fn(y_true, y_pred) do
    Axon.Losses.categorical_cross_entropy(y_true, y_pred,
      from_logits: true,
      sparse: true,
      reduction: :mean
    )
  end
end

{init_fn, predict_fn} = Axon.build(lora_model, mode: :train)
custom_predict_fn = &CommonTrain.custom_predict_fn(predict_fn, &1, &2)
custom_loss_fn = &CommonTrain.custom_loss_fn(&1, &2)

lora_merged_params =
  {init_fn, custom_predict_fn}
  |> Axon.Loop.trainer(custom_loss_fn, Polaris.Optimizers.adam(learning_rate: 3.0e-4))
  |> Axon.Loop.run(train_batch_stream, params, epochs: 1, iterations: 300, compiler: EXLA)

:ok

17:42:55.140 [debug] Forwarding options: [compiler: EXLA] to JIT compiler
Epoch: 0, Batch: 250, loss: 3.6697373
:ok

  
    
  
  Download LoRA params


When training, Axon will return all the parameters needed to run the model. If you want to download the LoRA-only parameters, you'll need to filter for them first before downloading.
# Method #1
lora_only =
  lora_merged_params
  |> Lorax.Params.filter(params)
  |> Lorax.Params.kino_download()

# Method #2 
# serialized =
#   Lorax.Params.filter(lora_merged_params, params)
#   |> Nx.serialize()

# File.write!("/<insert path>/test.lorax", serialized)

  
    
  
  Testing out text Generation


lora_model_info = %{model: lora_model, params: lora_merged_params, spec: spec}

lora_generation_config =
  Bumblebee.configure(generation_config,
    max_new_tokens: 256,
    strategy: %{type: :multinomial_sampling, top_p: 0.8}
  )

serving =
  Bumblebee.Text.generation(lora_model_info, tokenizer, lora_generation_config,
    compile: [batch_size: 1, sequence_length: 256],
    stream: true,
    defn_options: [compiler: EXLA, lazy_transfers: :always]
  )

Kino.start_child({Nx.Serving, name: Llama, serving: serving})
{:ok, #PID<0.1392.0>}
We'll kickstart the text generation using some <title>text</title> string. This is how the training data is formatted and will trigger the LoRA neurons to activate. With only 300 iterations, the model has already learned how to output text similar to our training data
Nx.Serving.batched_run(Llama, "<title>Elixir 2.0 released") |> Enum.each(&IO.write/1)
</title>

</likes>0 likes</likes>

<author>jlarperendan</author>

The elixir.clipper gem is really very elegant, you can even use it with other compilers if you don't need it in your code base. It really helps here!
<author>ilax13</author>

The name of the library is good! It already provides a lot of benefit from Elixir, so I guess it's useful for intermediate project developers to have one with both Elixir and Elixir 2.0.1.

<likes>0 likes</likes>

<author>benaeutz</author>

Elixir 2.0 is really powerful. Even though there's some backend changes, Elixir 2.0 is one of the few that changes the behavior of main mode and does not change runtime dependencies. So, if you have multiple compilers that can add the language layer or break the runtime dependencies, then the big advantage of Elixir is the compiler ecosystem. In many cases, the exception to this is compiler breaking.
<likes>1 likes</likes>

<author>avnicom<likes>1 likes
:ok



  

    
Running LLMs with LoRA
    

Mix.install([
  {:bumblebee, "~> 0.4.2"},
  {:axon, "~> 0.6.0"},
  {:nx, "~> 0.6.1"},
  {:exla, "~> 0.6.1"},
  {:explorer, "~> 0.7.0"},
  {:lorax, "~> 0.1.0"},
  {:req, "~> 0.4.0"},
  {:kino, "~> 0.11.0"}
])

Nx.default_backend(EXLA.Backend)

  
    
  
  Introduction


This notebook demonstrates how to run a text-generating model like GPT2 with LoRA. The basic steps:
	Define the model as it was during fine-tuning
	Merge the LoRA parameters with the base model weights.
	Run the model

We'll be using a LoRA file that was trained on the Elixir Chat/Discussion section. The data can be found here.

  
    
  
  Load original model


{:ok, spec} = Bumblebee.load_spec({:hf, "gpt2"})
{:ok, model} = Bumblebee.load_model({:hf, "gpt2"}, spec: spec)
{:ok, tokenizer} = Bumblebee.load_tokenizer({:hf, "gpt2"})
{:ok, generation_config} = Bumblebee.load_generation_config({:hf, "gpt2"})

%{model: model, params: gpt2_params} = model

:ok
:ok

  
    
  
  Define LoRA Model


Inject LoRA layers as it was during fine-tuning to ensure correct model compilation.
r = 4
lora_alpha = 8
lora_dropout = 0.05

lora_model =
  model
  |> Axon.freeze()
  |> Lorax.inject(%Lorax.Config{
    r: r,
    alpha: lora_alpha,
    dropout: lora_dropout,
    target_key: true,
    target_query: true,
    target_value: true
  })
#Axon<
  inputs: %{"attention_head_mask" => {12, 12}, "attention_mask" => {nil, nil}, "cache" => nil, "input_embeddings" => {nil, nil, 768}, "input_ids" => {nil, nil}, "position_ids" => {nil, nil}}
  outputs: "container_37"
  nodes: 895
>

  
    
  
  Load LoRA Params


There's 3 main ways to load your LoRA params
	URL: Use whatever HTTP library to retrieve the binary, and Nx.deserialize to get the map of tensor values
	File: Similar flow as URL. use File.read to retrieve the binary, Nx.deserialize. For convenience, you can use Lorax.Params.file_load!()
	Kino: GUI file picker. This requires a two-step process. You need to have one Kino input cell, and another cell to read the file input.

See the code below for examples
# Method 1: URL downlaod 
lora_serialized =
  Req.get!("https://raw.githubusercontent.com/wtedw/lorax/main/params/elixir-chat-r4a8.lorax").body
lora_only_params = Nx.deserialize(lora_serialized)
lora_only_params |> Map.keys()

# Method 2: File download
# lora_only_params = Lorax.Params.file_load!("<insert local path>")

# Method 3: Kino
# cell #1 
# input = Kino.Input.file("Lorax Params")
#
# cell #2 
# lora_only_params = Lorax.Params.kino_file_load!(input)
# merged_params = Lorax.Params.merge_params(lora_only_params, gpt2_params)
["dropout_28", "decoder.blocks.5.self_attention_dropout", "lora_11", "lora_31", "lora_10",
 "decoder.blocks.11.self_attention_dropout", "dropout_9", "lora_28", "dropout_25",
 "decoder.blocks.1.self_attention_dropout", "lora_13", "lora_20", "lora_7", "dropout_6", "lora_19",
 "decoder.blocks.2.self_attention_dropout", "lora_5", "lora_30",
 "decoder.blocks.7.self_attention_dropout", "lora_2", "lora_24", "dropout_1", "lora_6",
 "dropout_10", "dropout_3", "dropout_15", "lora_27", "dropout_7", "lora_35", "lora_22",
 "dropout_21", "decoder.blocks.3.self_attention_dropout", "lora_0", "dropout_27", "dropout_24",
 "lora_9", "decoder.blocks.4.self_attention_dropout", "lora_32", "dropout_34",
 "decoder.blocks.10.self_attention_dropout", "lora_23", "dropout_0",
 "decoder.blocks.6.self_attention_dropout", "lora_12", "dropout_16", "lora_34", "dropout_19",
 "lora_18", "lora_3", "lora_17", ...]

  
    
  
  Merge Params


Axon expects one single map of all the parameter values. So although we've loaded the LoRA params, we need to merge them with the original parameters.
merged_params = Map.merge(gpt2_params, lora_only_params)

:ok
:ok

  
    
  
  Inference Prepwork


The sampling method used here is non-deterministic, so the output is different everytime you run this cell + the bottom cell. Using top_p value of 0.6 may generate more coherent sentences, but tends to repeat. 0.7 and 0.8 is a good sweet spot.
lora_model_info = %{model: lora_model, params: merged_params, spec: spec}

lora_generation_config =
  Bumblebee.configure(generation_config,
    max_new_tokens: 512,
    strategy: %{type: :multinomial_sampling, top_p: 0.7}
  )

serving =
  Bumblebee.Text.generation(lora_model_info, tokenizer, lora_generation_config,
    compile: [batch_size: 1, sequence_length: 512],
    stream: true,
    defn_options: [compiler: EXLA, lazy_transfers: :always]
  )

Kino.start_child({Nx.Serving, name: Llama, serving: serving})
{:ok, #PID<0.9144.0>}

  
    
  
  Inference Results


The training data is formatted like this:
<title>Ideas for an Open Source Discord</title>

<author>WildYorkies</author>

I remember seeing on the Elixir subreddit [...]

<likes>3 likes</likes>
You can simulate a thread by kickstarting the text generation with some <title>text</title> string, and the LLM will give a prediction of what the general Elixir community will say.
Nx.Serving.batched_run(
  Llama,
  "<title>Elixir 2.0 is released! New features include</title>"
)
|> Enum.each(&IO.write/1)


<author>xjalasz</author>

Elixir 2.0 is released! New features include
This means that you can now deploy your Elixir 2 projects without having to use a tool like docker.

<likes>1 like</likes>

<author>jake</author>

As always, thanks for the help.
I am on a 10 day cruise to Toronto in July with the goal of finishing up Elixir 2.0 in less than one week.

<likes>1 like</likes>

<author>pianotato</author>

Thanks for all the support!

<likes>0 likes</likes>

<author>thekali</author>

Thanks for your time and effort.

<likes>0 likes</likes>

<author>mangai</author>

Thank you! I have no idea how long you have been on this blog.

<likes>1 like</likes>

<author>juicy_champion</author>

Thanks!

<likes>0 likes</likes>

<author>haha<likes>1 like</likes>

<author>thompson3</author>

Thank you for the support!

<likes>0 likes</likes>

<author>phrobotics</author>

Thanks for your time and effort.

<likes>0 likes</likes>

<author>yelp</author>

Thanks!

<likes>0 likes</likes>

<author>phrobotics</author>

Thanks!

<likes>0 likes</likes>

<author>jake</author>

Thanks for your time and effort.

<likes>0 likes</likes>

<author>jake</author>

Thanks!

<likes>0 likes</likes>

<author>juicy_champion</author>

Thanks!

<likes>0 likes</likes>

<author>juicy_champion</author>

Thanks!

<likes>0 likes</likes>

<author>gw
:ok



  

    
Stable Diffusion w/ LCM LoRA
    

Mix.install(
  [
    {:bumblebee, git: "https://github.com/elixir-nx/bumblebee.git"},
    {:nx, "~> 0.6.1", override: true},
    {:exla, "~> 0.6.1"},
    {:kino, "~> 0.11.0"},
    {:lorax, git: "https://github.com/wtedw/lorax.git"},
    {:axon, [env: :prod, git: "https://github.com/elixir-nx/axon.git", override: true]},
    {:req, "~> 0.4.0"}
  ],
  config: [nx: [default_backend: EXLA.Backend]]
)

Nx.global_default_backend(EXLA.Backend)

  
    
  
  Load SD1.5, LCM LoRA, and LCM scheduler


LCM LoRA requires two key things
	Loading the LoRA adapter for the appropriate Stable Diffusion unet model. We'll be using the LoRA file for SD 1.5
	LCMScheduler. The adapter will not work if any other scheduler is used.

We'll first download the safetensors from HF and call the Lorax library to convert the parameters to something Axon can use. In addition to converting the params, Lorax will provide a config that describes how new parameters should be injected into the Axon model.
repo_id = "runwayml/stable-diffusion-v1-5"
opts = [params_variant: "fp16"]

{:ok, tokenizer} = Bumblebee.load_tokenizer({:hf, "openai/clip-vit-large-patch14"})
{:ok, clip} = Bumblebee.load_model({:hf, repo_id, subdir: "text_encoder"}, opts)
{:ok, unet} = Bumblebee.load_model({:hf, repo_id, subdir: "unet"}, opts)
{:ok, vae} = Bumblebee.load_model({:hf, repo_id, subdir: "vae"}, [architecture: :decoder] ++ opts)
{:ok, scheduler} = Bumblebee.load_scheduler({:hf, repo_id, subdir: "scheduler"})
{:ok, featurizer} = Bumblebee.load_featurizer({:hf, repo_id, subdir: "feature_extractor"})
{:ok, safety_checker} = Bumblebee.load_model({:hf, repo_id, subdir: "safety_checker"}, opts)

# Option #1, Download LCM LoRA from HF
resp =
  Req.get!(
    "https://huggingface.co/latent-consistency/lcm-lora-sdv1-5/resolve/main/pytorch_lora_weights.safetensors?download=true"
  )

param_data = resp.body
{config, lcm_lora_params} = Lorax.Lcm.load!(param_data)

# Option #2, Load Locally
# {config, lcm_lora_params} =
#   Lorax.Lcm.read!("/Users/[user]/.../pytorch_lora_weights.safetensors")

# Axon expects one map containing all the layer names -> tensors. 
# We'll merge the LCM params with the original SD params.
lcm_unet = %{
  unet
  | model: Lorax.inject(unet.model, config),
    params: Map.merge(lcm_lora_params, unet.params)
}

lcm_scheduler = %Bumblebee.Diffusion.LcmScheduler{}

  
    
  
  Create serving + Kino prompts


serving =
  Bumblebee.Diffusion.StableDiffusion.text_to_image(clip, lcm_unet, vae, tokenizer, lcm_scheduler,
    num_steps: 4,
    num_images_per_prompt: 1,
    safety_checker_featurizer: featurizer,
    guidance_scale: 1.0,
    compile: [batch_size: 1, sequence_length: 60],
    defn_options: [compiler: EXLA]
  )

# compare this with regular SD 1.5
# serving =
#   Bumblebee.Diffusion.StableDiffusion.text_to_image(clip, unet, vae, tokenizer, scheduler,
#     num_steps: 4,
#     num_images_per_prompt: 1,
#     safety_checker_featurizer: featurizer,
#     compile: [batch_size: 1, sequence_length: 60],
#     defn_options: [compiler: EXLA]
#   )

prompt_input =
  Kino.Input.text("Prompt",
    default: "astronaut in the desert, high quality, detailed"
  )

negative_prompt_input = Kino.Input.text("Negative Prompt", default: "blurry")

Kino.Layout.grid([prompt_input, negative_prompt_input])

  
    
  
  Text to Image


some notes about generation
	Although LCM LoRA speeds up image generation, it's sensitive to "correct" prompts.
	Some random seeds produce good images, some are complete gibberish (especially if you only have one word like "turtle" as your prompt).
	The LCM folks say 2-8 steps is a good range, but I found that 4 steps and guidance_scale = 1.0 is best. 8 steps causes too much denoising to happen, so the images appear more saturated.

We are ready to generate images!
prompt = Kino.Input.read(prompt_input)
negative_prompt = Kino.Input.read(negative_prompt_input)
output = Nx.Serving.run(serving, %{prompt: prompt, negative_prompt: negative_prompt})

for result <- output.results do
  Kino.Image.new(result.image)
end
|> Kino.Layout.grid(columns: 2)



  

    
Stable Diffusion w/ LCM (Realtime)
    

Mix.install(
  [
    {:bumblebee, git: "https://github.com/elixir-nx/bumblebee.git"},
    {:nx, "~> 0.6.1", override: true},
    {:exla, "~> 0.6.1"},
    {:kino, "~> 0.11.0"},
    {:lorax, git: "https://github.com/wtedw/lorax.git"},
    {:axon, [env: :prod, git: "https://github.com/elixir-nx/axon.git", override: true]},
    {:req, "~> 0.4.0"}
  ],
  config: [nx: [default_backend: EXLA.Backend]]
)

Nx.global_default_backend(EXLA.Backend)

  
    
  
  Load SD1.5, LCM LoRA, and LCM scheduler


LCM LoRA requires two key things
	Loading the LoRA adapter for the appropriate Stable Diffusion unet model. We'll be using the LoRA file for SD 1.5
	LCMScheduler. The adapter will not work if any other scheduler is used.

We'll first download the safetensors from HF and call the Lorax library to convert the parameters to something Axon can use. In addition to converting the params, Lorax will provide a config that describes how new parameters should be injected into the Axon model.
repo_id = "runwayml/stable-diffusion-v1-5"
opts = [params_variant: "fp16"]

{:ok, tokenizer} = Bumblebee.load_tokenizer({:hf, "openai/clip-vit-large-patch14"})
{:ok, clip} = Bumblebee.load_model({:hf, repo_id, subdir: "text_encoder"}, opts)
{:ok, unet} = Bumblebee.load_model({:hf, repo_id, subdir: "unet"}, opts)
{:ok, vae} = Bumblebee.load_model({:hf, repo_id, subdir: "vae"}, [architecture: :decoder] ++ opts)
{:ok, scheduler} = Bumblebee.load_scheduler({:hf, repo_id, subdir: "scheduler"})
{:ok, featurizer} = Bumblebee.load_featurizer({:hf, repo_id, subdir: "feature_extractor"})
{:ok, safety_checker} = Bumblebee.load_model({:hf, repo_id, subdir: "safety_checker"}, opts)

# Option #1, Download LCM LoRA from HF
resp =
  Req.get!(
    "https://huggingface.co/latent-consistency/lcm-lora-sdv1-5/resolve/main/pytorch_lora_weights.safetensors?download=true"
  )

param_data = resp.body
{config, lcm_lora_params} = Lorax.Lcm.load!(param_data)

# Option #2, Load Locally
# {config, lcm_lora_params} =
#   Lorax.Lcm.read!("/Users/[user]/.../pytorch_lora_weights.safetensors")

# Axon expects one map containing all the layer names -> tensors. 
# We'll merge the LCM params with the original SD params.
lcm_unet = %{
  unet
  | model: Lorax.inject(unet.model, config),
    params: Map.merge(lcm_lora_params, unet.params)
}

lcm_scheduler = %Bumblebee.Diffusion.LcmScheduler{}

  
    
  
  Create serving


serving =
  Bumblebee.Diffusion.StableDiffusion.text_to_image(clip, lcm_unet, vae, tokenizer, lcm_scheduler,
    num_steps: 4,
    num_images_per_prompt: 1,
    safety_checker_featurizer: featurizer,
    guidance_scale: 1.0,
    compile: [batch_size: 1, sequence_length: 60],
    defn_options: [compiler: EXLA]
  )

  
    
  
  Realtime LCM LoRA


Run all the following cells and begin typing in the prompt below. The Kino frame beneath the prompt should reflect new images. Debounce value can be changed to control how fast the images should refresh
This has been tested on the NVIDIA A10G and above.
Make sure you use a high end graphics card for this or else the frame below will be blank for quite a long time.
input = Kino.Input.textarea("Prompt", debounce: 150)
# Images will be rendered in this frame
frame = Kino.Frame.new()
Kino.listen(input, fn %{value: prompt} ->
  # Generate the image and put it in the frame
  output = Nx.Serving.run(serving, %{prompt: prompt})
  [result] = output.results
  image = Kino.Image.new(result.image)
  Kino.Frame.render(frame, image)
end)



  

    
Lorax 
    



      
Simple Low-Rank Adaptation (LoRA) implementation

  
    
  
  LoRA model creation


To create a LoRA model, freeze an existing model and inject LoRA layers using Lorax.inject/2.
lora_model =
  model
  |> Axon.freeze()
  |> Lorax.inject(%Lorax.Config{
    r: 2,
    alpha: 4,
    dropout: 0.05,
    target_key: false,
    target_query: false,
    target_value: true
  })
For more detailed guides, see
	Finetuning LLMs with LoRA
	Running LLMs with LoRA

LoRA layers are implemented by injecting new nodes into the Axon struct.
These LoRA nodes represent the B and A matrices. Each node takes an input x and computes BAx.
Furthermore, the LoRA node will receive Wx as an input and compute Wx + BAx.
This isn't the standard implementation, but it simplifies the injection process.

  
    
  
  Injection Process


Beginning state
flowchart LR
  A[input id:0] --> B[target id:1]
Create an empty dummy node
flowchart LR
  A[input id:0] --> B[target id:1] --> C[dummy id:2]
Create lora node with input ids = [0, 2]
flowchart LR
  A[input id:0] --> B[target id:1] --> C[dummy id:2] --> E[lora id:3]
  A[input id:0] --> E[lora id:3]
target takes dummy's id, throw away dummy node
flowchart LR
  A[input id:0] --> C[target id:2]
  C[target id:2] --> E[lora id:3]
  A[input id:0] --> E[lora id:3]
lora takes target's original id
flowchart LR
  A[input id:0] --> C[target id:2] --> E[lora id:1]
  A[input id:0] --> E[lora id:1]
lora and target are now swapped.
Any downstream node that relied on node id:1 will now receive Wx + BAx

      


      
        Summary


  
    Functions
  


    
      
        inject(axon, config)

      


        Returns a modified Axon model with LoRA nodes inserted according to the provided configuration.



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    inject(axon, config)



  


  

Returns a modified Axon model with LoRA nodes inserted according to the provided configuration.
target_key, target_query, target_value are required if target_node_fn isn't specified

  
    
  
  Examples


lora_model =
  model
  |> Axon.freeze()
  |> Lorax.inject(%Lorax.Config{
    r: 2,
    alpha: 4,
    dropout: 0.05,
    target_key: true,
    target_query: true,
    target_value: true
  })

  
    
  
  Targeting nodes manually


lora_model =
  model
  |> Axon.freeze()
  |> Lorax.inject(%Lorax.Config{
    r: 2,
    alpha: 4,
    dropout: 0.05,
    target_node_fn: fn %Axon.Node{name: name_fn} ->
      # names are generated lazily, and look like "decoder.blocks.11.self_attention.value"
      # have to invoke the function to see what layer the node represents
      # https://github.com/elixir-nx/axon/blob/v0.6.0/lib/axon.ex#L3923
      name = name_fn.(nil, nil)
      shortname = String.split(name, ".") |> List.last()

      if shortname == "output" do
        true
      else
        false
      end
    end
  })

  


        

      



  

    
Lorax.Config 
    



      
Config for Lorax.inject/2
r is the rank in the low-rank matrices used in LoRA.
A higher value of r increases the expressiveness of the adaptation,
However, it also increases the number of parameters and the computational
cost. Conversely, a lower value of r makes the adaptation simpler and less
resource-intensive. Defaults to 1.
alpha is a scaling factor that controls the magnitude of changes introduced
by the low-rank matrices. A higher value of alpha means that the
modifications made by LoRA have a greater impact on the model's original
weights. This can lead to more significant changes in the model's behavior.
A lower value results in more subtle changes. Defaults to 2.
dropout specifies the dropout rate applied to the low-rank matrices.
dropout_seed determines the seed used for Nx.Random.key/1 during
dropout application. When defined, it ensures that the LoRA adapter
produces consistent tensor values, assuming that other layers also have
deterministic outputs.
param_type specifies the numerical representation for the A and B
matrices. Defaults to float32
target_query specifies whether to apply LoRA to all query matrices in an
attention block. Defaults to true.
target_value specifies whether to apply LoRA to all value matrices in an
attention block. Defaults to true.
target_key specifies whether to apply LoRA to all key matrices in an
attention block. Defaults to true.

      





  

    
Lorax.Lcm 
    




      
        Summary


  
    Functions
  


    
      
        load!(data)

      


        Loads some serialized LCM LoRA data and returns a returns a tuple of {%Lorax.Config{}, lora_params}



    


    
      
        read!(path)

      


        Reads some LCM LoRA safetensor file and returns a tuple of {%Lorax.Config{}, lora_params}



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    load!(data)



  


  

Loads some serialized LCM LoRA data and returns a returns a tuple of {%Lorax.Config{}, lora_params}

  



  
    
      
      Link to this function
    
    read!(path)



  


  

Reads some LCM LoRA safetensor file and returns a tuple of {%Lorax.Config{}, lora_params}

  


        

      



  

    
Lorax.Params 
    



      
Helper module for loading, downloading, filtering, and calculating the size of Axon parameters

      


      
        Summary


  
    Functions
  


    
      
        file_load!(params_path)

      


        Loads parameters from file path



    


    
      
        filter(lora_merged_params, original_params)

      


        Returns LoRA only params from a merged param map



    


    
      
        kino_download(params, filename \\ "params.lorax", label \\ "Download Params")

      


        Creates a Kino widget for downloading params map.
Must be placed in the last line of a Livebook cell.



    


    
      
        kino_file_load!(kino_input)

      


        Creates a Kino widget for uploading serialized params file.
Must be placed in the last line of a Livebook cell.



    


    
      
        size(params)

      


        Calculates total bytes of all the tensors inside a parameter map



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    file_load!(params_path)



  


  

Loads parameters from file path

  



  
    
      
      Link to this function
    
    filter(lora_merged_params, original_params)



  


  

Returns LoRA only params from a merged param map

  



    

    

  
    
      
      Link to this function
    
    kino_download(params, filename \\ "params.lorax", label \\ "Download Params")



  


  

Creates a Kino widget for downloading params map.
Must be placed in the last line of a Livebook cell.

  



  
    
      
      Link to this function
    
    kino_file_load!(kino_input)



  


  

Creates a Kino widget for uploading serialized params file.
Must be placed in the last line of a Livebook cell.

  



  
    
      
      Link to this function
    
    size(params)



  


  

Calculates total bytes of all the tensors inside a parameter map

  


        

      



  

    
Lorax.Shape 
    




      
        Summary


  
    Functions
  


    
      
        calc_ab(op, r, parameters)

      


        Infers the shape of the LoRA matrices. Only supports :dense and :conv operations



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    calc_ab(op, r, parameters)



  


  

Infers the shape of the LoRA matrices. Only supports :dense and :conv operations

  
    
  
  Dense LoRA kernels


Suppose we have a target node W and input tensor x,
During injection, the A matrix will project the input tensor down to
r-dimensional space. Afterwards, the B matrix will project the result back up
to some unknown dimensionality. To figure out the output dimensionality,
we inspect W's kernel shape.

  
    
  
  Convolution LoRA kernels


In addition to figuring out the output dimensionality, we need to retrieve
the convolution kernel_size of W. When calling Axon.conv, it's passed as an
option, but then is no longer stored inside the Axon node. To figure out the
kernel size, we also inspect W's kernel shape.

